
Context

From knowledge to action

From theoretical to practical reasoning

From mind to body & environment & others (interaction)

From individuals to societies

Context

An MAS is a loosely coupled network of problem solvers
that interact to solve problems that are beyond the indi-
vidual capabilities or knowledge of each problem solver

– Durfee and Lesser 1989

Outline

▶ Agents
▶ Practical reasoning
▶ Jason

▶ Environment
▶ Organisation
▶ MAOP

(slides written together with R. Bordini, O. Boissier, and A. Ricci)

Agent Oriented
Programming
— AOP —



environment

?



Literature

Books: [Bordini et al., 2005], [Bordini et al., 2009]

Proceedings: ProMAS, DALT, LADS, EMAS, AGERE, ...

Surveys: [Bordini et al., 2006], [Fisher et al., 2007] ...

Languages of historical importance: Agent0 [Shoham, 1993],
AgentSpeak(L) [Rao, 1996], MetateM [Fisher, 2005],
3APL [Hindriks et al., 1997],
Golog [Giacomo et al., 2000]

Other prominent languages:
Jason [Bordini et al., 2007],
Jadex [Pokahr et al., 2005], 2APL [Dastani, 2008],
GOAL [Hindriks, 2009], JACK [Winikoff, 2005],
JIAC, ASTRA

But many others languages and platforms...

Some Languages and Platforms

Jason (Hübner, Bordini, ...); 3APL and 2APL (Dastani, van
Riemsdijk, Meyer, Hindriks, ...); Jadex (Braubach, Pokahr);
MetateM (Fisher, Guidini, Hirsch, ...); ConGoLog (Lesperance,
Levesque, ... / Boutilier – DTGolog); Teamcore/ MTDP (Milind
Tambe, ...); IMPACT (Subrahmanian, Kraus, Dix, Eiter); CLAIM
(Amal El Fallah-Seghrouchni, ...); GOAL (Hindriks); BRAHMS
(Sierhuis, ...); SemantiCore (Blois, ...); STAPLE (Kumar, Cohen,
Huber); Go! (Clark, McCabe); Bach (John Lloyd, ...); MINERVA
(Leite, ...); SOCS (Torroni, Stathis, Toni, ...); FLUX (Thielscher);
JIAC (Hirsch, ...); JADE (Agostino Poggi, ...); JACK (AOS); Agentis
(Agentis Software); Jackdaw (Calico Jack); ASTRA (Rem Collier);
SARL (Stephane Galland); simpAL, ALOO (Ricci, ...);...

Agent Oriented Programming
Features

▶ Reacting to events × long-term goals
▶ Course of actions depends on circumstance
▶ Plan failure (dynamic environments)
▶ Social ability
▶ Combination of theoretical and practical reasoning

Agent Oriented Programming
Fundamentals

▶ Use of mentalistic notions and a societal view of
computation [Shoham, 1993]

▶ Heavily influenced by the BDI architecture and reactive
planning systems [Bratman et al., 1988]

Motivation for BDI — autonomous robot
[Cohen and Levesque, 1990]

BDI architecture
(the mentalistic view)

Beliefs

Desires

Intentions means-end

deliberate

BRF

options

perception

action

BDI architecture [Wooldridge, 2009]

1 while true do
2 B ← br f (B, perception()) // belief revision
3 D ← options(B, I) // desire revision
4 I ← del iberate(B,D, I) // get intentions
5 π ← meansend(B, I, A) // gets a plan
6 while π ̸= ∅ do
7 execute( head(π) )
8 π ← tai l(π)

BDI architecture [Wooldridge, 2009]

1 while true do
2 B ← br f (B, perception()) // belief revision
3 D ← options(B, I) // desire revision
4 I ← del iberate(B,D, I) // get intentions
5 π ← meansend(B, I, A) // gets a plan
6 while π ̸= ∅ do
7 execute( head(π) )
8 π ← tai l(π)

fine for pro-activity, but not for reactivity (over commitment)

BDI architecture [Wooldridge, 2009]

1 while true do
2 B ← br f (B, perception()) // belief revision
3 D ← options(B, I) // desire revision
4 I ← del iberate(B,D, I) // get intentions
5 π ← meansend(B, I, A) // gets a plan
6 while π ̸= ∅ do
7 execute( head(π) )
8 π ← tai l(π)
9 B ← br f (B, perception())

10 if ¬sound(π, I, B) then
11 π ← meansend(B, I, A)

revise commitment to plan – re-planning for context adaptation

BDI architecture [Wooldridge, 2009]

1 while true do
2 B ← br f (B, perception()) // belief revision
3 D ← options(B, I) // desire revision
4 I ← del iberate(B,D, I) // get intentions
5 π ← meansend(B, I, A) // gets a plan
6 while π ̸= ∅ and ¬succeeded(I, B) and ¬impossible(I, B) do
7 execute( head(π) )
8 π ← tai l(π)
9 B ← br f (B, perception())

10 if ¬sound(π, I, B) then
11 π ← meansend(B, I, A)

revise commitment to intentions – Single-Minded Commitment

BDI architecture [Wooldridge, 2009]

1 while true do
2 B ← br f (B, perception()) // belief revision
3 D ← options(B, I) // desire revision
4 I ← del iberate(B,D, I) // get intentions
5 π ← meansend(B, I, A) // gets a plan
6 while π ̸= ∅ and ¬succeeded(I, B) and ¬impossible(I, B) do
7 execute( head(π) )
8 π ← tai l(π)
9 B ← br f (B, perception())

10 if reconsider(I, B) then
11 D ← options(B, I)
12 I ← del iberation(B,D, I)
13 if ¬sound(π, I, B) then
14 π ← meansend(B, I, A)

reconsider the intentions (not always!)
13



Jason
(let’s go programming those nice concepts)



(BDI & Jason) Hello World – agent bob

happy(bob). // B

!say(hello). // D

+!say(X) : happy(bob) // I

<- .print(X).

(BDI & Jason) Hello World – agent bob

happy(bob). // B

!say(hello). // D

+!say(X) : happy(bob) // I

<- .print(X).

15

beliefs

▶ prolog like (FOL)



(BDI & Jason) Hello World – agent bob

happy(bob). // B

!say(hello). // D

+!say(X) : happy(bob) // I

<- .print(X).

15

desires

▶ prolog like
▶ with ! prefix



(BDI & Jason) Hello World – agent bob

happy(bob). // B

!say(hello). // D

+!say(X) : happy(bob) // I

<- .print(X).

15

plans

▶ define when a desire
becomes an intention
⇝ deliberate

▶ how it is satisfied
▶ are used for practical

reasoning
⇝ means-end



Hello World
desires from perception — options

+happy(bob) <- !say(hello).

+!say(X) : not today(monday)
<- .print(X).

Hello World
source of beliefs

+happy(bob)[source(A)]
: someone_who_knows_me_very_well(A)
<- !say(hello).

+!say(X) : not today(monday) <- .print(X).

Hello World
plan selection

+happy(H)[source(A)]
: sincere(A) & .my_name(H)
<- !say(hello).

+happy(H)
: not .my_name(H)
<- !say(i_envy(H)).

+!say(X) : not today(monday) <- .print(X).

Hello World
intention revision

+happy(H)[source(A)]
: sincere(A) & .my_name(H)
<- !say(hello).

+happy(H)
: not .my_name(H)
<- !say(i_envy(H)).

+!say(X) : not today(monday) <- .print(X); !say(X).

-happy(H)
: .my_name(H)
<- .drop_intention(say(hello)).

19



Hello World
intention revision

+happy(H)[source(A)]
: sincere(A) & .my_name(H)
<- !say(hello).

+happy(H)
: not .my_name(H)
<- !say(i_envy(H)).

+!say(X) : not today(monday) <- .print(X); !say(X).

-happy(H)
: .my_name(H)
<- .drop_intention(say(hello)).

Hello World
intention revision

+happy(H)[source(A)]
: sincere(A) & .my_name(H)
<- !say(hello).

+happy(H)
: not .my_name(H)
<- !say(i_envy(H)).

+!say(X) : not today(monday) <- .print(X); !say(X).

-happy(H)
: .my_name(H)
<- .drop_intention(say(hello)).

features

▶ we can have several intentions
based on the same plans
⇝ running concurrently

▶ long term goals running
⇝ reaction meanwhile
⇝ not overcommitted

▶ plan selection based on
circumstance

▶ actions (partially) computed by
the interpreter
⇝ programmer declares plans



AgentSpeak
The foundational language for Jason

▶ Originally proposed by Rao [Rao, 1996]
▶ Programming language for BDI agents
▶ Elegant notation, based on logic programming
▶ Inspired by PRS (Georgeff & Lansky), dMARS (Kinny), and

BDI Logics (Rao & Georgeff)
▶ Abstract programming language aimed at theoretical results

Jason
A practical implementation of a variant of AgentSpeak

▶ Jason implements the operational semantics of a variant of
AgentSpeak

▶ Has various extensions aimed at a more practical
programming language (e.g. definition of the MAS,
communication, ...)

▶ Highly customised to simplify extension and
experimentation

▶ Developed by Jomi F. Hübner, Rafael H. Bordini, and others

Main Language Constructs

Beliefs: represent the information available to an agent (e.g.
about the environment or other agents)

Goals: represent states of affairs the agent wants to bring
about

Plans: are recipes for action, representing the agent’s
know-how

Beliefs — Representation

Syntax

Beliefs are represented by annotated literals of first order logic

functor(term1, …, termn)[annot1, …, annotm]

Example (belief base of agent Tom)

red(box1)[source(percept)].
friend(bob,alice)[source(bob)].
lier(alice)[source(self),source(bob)].
~lier(bob)[source(self)].

Beliefs — Dynamics I

by perception

beliefs annotated with source(percept) are automatically updated
accordingly to the perception of the agent

by intention

the plan operators + and - can be used to add and remove beliefs
annotated with source(self) (mental notes)

+lier(alice); // adds lier(alice)[source(self)]
-lier(john); // removes lier(john)[source(self)]

24



Beliefs — Dynamics II

by communication

when an agent receives a tell message, the content is a new
belief annotated with the sender of the message

.send(tom,tell,lier(alice)); // sent by bob
// adds lier(alice)[source(bob)] in Tom's BB
...
.send(tom,untell,lier(alice)); // sent by bob
// removes lier(alice)[source(bob)] from Tom's BB

Goals — Representation

Types of goals

▶ Achievement goal: goal to do
▶ Test goal: goal to know

Syntax

Goals have the same syntax as beliefs, but are prefixed by
! (achievement goal) or
? (test goal)

Example (Initial goal of agent Tom)

!write(book).

Goals — Dynamics I

by intention

the plan operators ! and ? can be used to add a new goal
annotated with source(self)

...
// adds new achievement goal !write(book)[source(self)]
!write(book);

// adds new test goal ?publisher(P)[source(self)]
?publisher(P);
...

27



Goals — Dynamics II

by communication – achievement goal

when an agent receives an achieve message, the content is a new
achievement goal annotated with the sender of the message

.send(tom,achieve,write(book)); // sent by Bob
// adds new goal write(book)[source(bob)] for Tom
...
.send(tom,unachieve,write(book)); // sent by Bob
// removes goal write(book)[source(bob)] for Tom

Goals — Dynamics III

by communication – test goal

when an agent receives an askOne or askAll message, the content
is a new test goal annotated with the sender of the message

.send(tom,askOne,published(P),Answer); // sent by Bob
// adds new goal ?publisher(P)[source(bob)] for Tom
// the response of Tom unifies with Answer

Triggering Events — Representation

▶ Events happen as consequence to changes in the agent’s
beliefs or goals

▶ An agent reacts to events by executing plans
▶ Types of plan triggering events

+b (belief addition)
-b (belief deletion)
+!g (achievement-goal addition)
-!g (achievement-goal deletion)
+?g (test-goal addition)
-?g (test-goal deletion)

Plans — Representation

An AgentSpeak plan has the following general structure:

triggering_event : context <- body.

where:
▶ the triggering event denotes the events that the plan is meant

to handle
▶ the context represent the circumstances in which the plan

can be used
▶ the body is the course of action to be used to handle the

event if the context is believed true at the time a plan is
being chosen to handle the event

31



Plans — Operators for Plan Context

Boolean operators

& (and)

| (or)
not (not)

= (unification)

>, >= (relational)

<, <= (relational)

== (equals)

\ == (different)

Arithmetic operators

+ (sum)

- (subtraction)

* (multiply)

/ (divide)

div (divide – integer)

mod (remainder)

** (power)

Plans — Operators for Plan Body

+rain : time_to_leave(T) & clock.now(H) & H >= T
<- !g1; // new sub-goal

!!g2; // new goal
?b(X); // new test goal
+b1(T-H); // add mental note
-b2(T-H); // remove mental note
-+b3(T*H); // update mental note
jia.get(X); // internal action
X > 10; // constraint to carry on
close(door);// external action
!g3[hard_deadline(3000)]. // goal with deadline

Plans — Example

+green_patch(Rock)[source(percept)]
: not battery_charge(low)
<- ?location(Rock,Coordinates);

!at(Coordinates);
!examine(Rock).

+!at(Coords)
: not at(Coords) & safe_path(Coords)
<- move_towards(Coords);

!at(Coords).
+!at(Coords)

: not at(Coords) & not safe_path(Coords)
<- ...

+!at(Coords) : at(Coords).

Plans — Dynamics

The plans that form the plan library of the agent come from
▶ initial plans defined by the programmer
▶ plans added dynamically and intentionally by

▶ .add_plan
▶ .remove_plan

▶ plans received from
▶ tellHow messages
▶ untellHow

Main Language Constructs and Runtime Structures

Beliefs: represent the information available to an agent (e.g.
about the environment or other agents)

Goals: represent states of affairs the agent wants to bring
about

Plans: are recipes for action, representing the agent’s
know-how

Events: happen as consequence to changes in the agent’s
beliefs or goals

Intentions: plans instantiated to achieve some goal

Basic Reasoning cycle
runtime interpreter

▶ perceive the environment and update belief base
▶ process new messages
▶ select event
▶ select relevant plans
▶ select applicable plans
▶ create/update intention
▶ select intention to execute
▶ execute one step of the selected intention

Jason Reasoning Cycle

▶ machine perception
▶ belief revison
▶ knowledge

representation
▶ communication,

argumentation
▶ trust
▶ social power



Jason Reasoning Cycle

SI

Events
External

Event
Selected

SE

Beliefs to
Add and

Delete

Relevant
Plans

New Plan
Push

Intention
Updated

OS

Applicable
Plans

Means
Intended

Events
External

Plan
Library

Events

Internal
Events

3

checkMail

Intentions

Execute
Intention

...New
New

9

Belief
Base

New
Intention

Percepts

act

Selected
Intention

Intentions

Action

Percepts
1 2

BUF

10

Events

Context
Check

Event
Unify

BRF

Beliefs

Agent

sendMsg

Beliefs

8

Messages

Plans

perceive

7

5

6

Actions

Beliefs

Suspended Intentions
(Actions and Msgs)

...

.send

SocAcc

4

Messages Messages
SM

40

▶ planning
▶ reasoning
▶ decision theoretic

techniques
▶ learning

(reinforcement)

▶ intention
reconsideration

▶ scheduling
▶ action theories



A note about “Control”

Agents can control (manipulate) their own (and influence the
others)
▶ beliefs
▶ goals
▶ plan

By doing so they control their behaviour

The developer provides initial values of these elements and thus
also influence the behaviour of the agent

Failure Handling: Contingency Plans

Example (an agent blindly committed to g)

+!g : g. // g is a declarative goal

+!g : ... <- a1; ?g.
+!g : ... <- a2; ?g.
+!g : ... <- a3; ?g.

+!g <- !g. // keep trying
-!g <- !g. // in case of some failure

+g <-.succeed_goal(g).

Failure Handling: Contingency Plans

Example (single minded commitment)

+!g : g. // g is a declarative goal

+!g : ... <- a1; ?g.
+!g : ... <- a2; ?g.
+!g : ... <- a3; ?g.

+!g <- !g. // keep trying
-!g <- !g. // in case of some failure

+g <-.succeed_goal(g).
+f : .super_goal(g,SG) <-.fail_goal(SG).
f is the drop condition for goal g

Compiler pre-processing – directives

Example (single minded commitment)

{ begin smc(g,f) }
+!g : ... <- a1.
+!g : ... <- a2.
+!g : ... <- a3.

{ end }

Meta Programming

Example (an agent that asks for plans on demand)

-!G[error(no_relevant)] : teacher(T)
<- .send(T, askHow, { +!G }, Plans);

.add_plan(Plans);
!G.

in the event of a failure to achieve any goal G due to no
relevant plan, asks a teacher for plans to achieve G and
then try G again

▶ The failure event is annotated with the error type, line,
source, ... error(no_relevant) means no plan in the agent’s
plan library to achieve G

▶ { +!G } is the syntax to enclose triggers/plans as terms

Other Language Features
Strong Negation

+!leave(home)
: ~raining
<- open(curtains); ...

+!leave(home)
: not raining & not ~raining
<- .send(mum,askOne,raining,Answer,3000); ...

Prolog-like Rules in the Belief Base

tall(X) :- woman(X) & height(X, H) & H > 1.70.
tall(X) :- man(X) & height(X, H) & H > 1.80.

Internal Actions

▶ Unlike actions, internal actions do not change the
environment

▶ Code to be executed as part of the agent reasoning cycle
▶ AgentSpeak is meant as a high-level language for the agent’s

practical reasoning and internal actions can be used for
invoking legacy code elegantly

▶ Internal actions can be defined by the user in Java

libname.action_name(. . .)

Standard Internal Actions

▶ Standard (pre-defined) internal actions have an empty library
name
▶ .print(term1, term2, . . .)
▶ .union(l i st1, l i st2, l i st3)
▶ .my_name(var)
▶ .send(ag,per f ,l i teral)
▶ .intend(l i teral)
▶ .drop_intention(l i teral)

▶ Many others available for: printing, sorting, list/string
operations, manipulating the beliefs/annotations/plan library,
creating agents, waiting/generating events, etc.

Namespaces & Modularity

50



Namespaces & Modularity

{include("initiator.asl", pc)}
{include("initiator.asl", tv)}

!pc::startCNP(fix(pc)).
!tv::startCNP(fix(tv)).

+pc::winner(X)
<- .print(X).

Jason Customisations

▶ Agent class customisation:
selectMessage, selectEvent, selectOption, selectIntention,
buf, brf, ...

▶ Agent architecture customisation:
perceive, act, sendMsg, checkMail, ...

▶ Belief base customisation:
add, remove, contains, ...
▶ Example available with Jason: persistent belief base (in text

files, in data bases, ...)

Jason × Java

Consider a very simple robot with two goals:
▶ when a piece of gold is seen, go to it
▶ when battery is low, go charge it

Java code – go to gold

public class Robot extends Thread {
boolean seeGold, lowBattery;
public void run() {

while (true) {
while (! seeGold) {

a = randomDirection();
doAction(go(a));

}
while (seeGold) {

a = selectDirection();

doAction(go(a));

} } } }

Java code – charge battery

public class Robot extends Thread {
boolean seeGold, lowBattery;
public void run() {

while (true) {
while (! seeGold) {

a = randomDirection();
doAction(go(a));
if (lowBattery) charge();

}
while (seeGold) {

a = selectDirection();
if (lowBattery) charge();
doAction(go(a));
if (lowBattery) charge();

} } } }

Jason code

direction(gold) :- see(gold).
direction(random) :- not see(gold).

+!find(gold) // long term goal
<- ?direction(A);

go(A);
!find(gold).

+battery(low) // reactivity
<- !charge.

^!charge[state(executing)] // goal meta-events
<- .suspend(find(gold)).

^!charge[state(finished)]
<- .resume(find(gold)).

Jason × Prolog

▶ With the Jason extensions, nice separation of theoretical and
practical reasoning

▶ BDI architecture allows
▶ long-term goals (goal-based behaviour)
▶ reacting to changes in a dynamic environment
▶ handling multiple foci of attention (concurrency)

▶ Acting on an environment and a higher-level conception of
a distributed system

Summary

▶ AgentSpeak
▶ Logic + BDI
▶ Agent programming language

▶ Jason
▶ AgentSpeak interpreter
▶ Implements the operational semantics of AgentSpeak
▶ Speech-act based communicaiton
▶ Highly customisable
▶ Useful tools
▶ Open source
▶ Open issues

Further Resources

▶ http://jason.sourceforge.net

▶ R.H. Bordini, J.F. Hübner, and
M. Wooldrige
Programming Multi-Agent Systems in
AgentSpeak using Jason
John Wiley & Sons, 2007.

Environment Oriented
Programming
— EOP —



?



Back to the Notion of Environment in MAS

▶ The notion of environment is intrinsically related to the
notion of agent and multi-agent system
▶ “An agent is a computer system that is situated in some

environment and that is capable of autonomous action in
this environment in order to meet its design
objective” [Wooldridge, 2002]

▶ “An agent is anything that can be viewed as perceiving its
environment through sensors and acting upon the
environment through effectors. ” [Russell and Norvig, 2003]

▶ Including both physical and software environments

Single Agent Perspective

ENVIRONMENT

feedback

actions

percepts
effectors / actuators

sensors

acti
on to

 d
o

PERCEPTION

DECISION

ACTION

▶ Perception
▶ process inside agent inside of attaining awareness or

understanding sensory information, creating percepts
perceived form of external stimuli or their absence

▶ Actions
▶ the means to affect, change or inspect the environment

Multi-Agent Perspective

▶ In evidence
▶ overlapping spheres of visibility and influence
▶ ..which means: interaction

Why Environment Programming

▶ Basic level
▶ to create testbeds for real/external environments
▶ to ease the interface/interaction with existing software

environments
▶ Advanced level

▶ to uniformly encapsulate and modularise functionalities of
the MAS out of the agents
▶ typically related to interaction, coordination, organisation,

security
▶ externalisation

▶ this implies changing the perspective on the environment
▶ environment as a first-class abstraction of the MAS
▶ endogenous environments (vs. exogenous ones)
▶ programmable environments

Basic Level Overview

actions

percepts

SIMULATED 

WORLD

OR

INTERFACE

OR 

WRAPPER TO

EXISTING 

TECHNOLOGY 

EXTERNAL 

WORLD
(PHYSICAL OR 

COMPUTATIONAL)

MAS  ENVIRONMENT

REAL WORLD
(PHYSICAL OR 

COMPUTATIONAL)

mimicking

Example:

JAVA 

PLATFORMAGENTS

MAS 

Advanced Level Overview [Weyns et al., 2007]

Existing Computational Frameworks

▶ AGRE / AGREEN / MASQ [Stratulat et al., 2009]
▶ AGRE – integrating the AGR (Agent-Group-Role) organisation

model with a notion of environment
▶ Environment used to represent both the physical and social

part of interaction
▶ AGREEN / MASQ – extending AGRE towards a unified

representation for physical, social and institutional
environments

▶ Based on MadKit platform [Gutknecht and Ferber, 2000]
▶ GOLEM [Bromuri and Stathis, 2008]

▶ Logic-based framework to represent environments for
situated cognitive agents

▶ composite structure containing the interaction between
cognitive agents and objects

▶ A&A and CArtAgO [Ricci et al., 2010a]
▶ introducing a computational notion of artifact to design and

implement agent environments

A&A and CArtAgO



Agents and Artifacts (A&A) Conceptual Model:
Background Human Metaphor

WHITEBOARD
artifact

ARCHIVE
artifact

COM. CHANNEL
artifact

TASK SCHEDULER
artifact

RESOURCE 
artifact

CLOCK
artifact

BAKERY

workspace

agents can join
dynamically the workspace

70



A&A Meta-Model in More Detail [Ricci et al., 2010a]

Artifact

Operation

Observable 
Event

generate

Agentuse

perceive

Workspace

Environment

Observable
Property

update

perceive

observe

Manual

has

consult

link

create

dispose

link

join

quit

Artifact Abstract Representation

OperationX(Params)

...

ObsPropName(Args)

...

SIGNALS

USAGE

INTERFACE

OBSERVABLE 

PROPERTIES

OperationY(Params)

...

LINK

INTERFACE

OPERATIONS 

A World of Artifacts

put

n_items 0

max_items 100

get

a bounded buffer

inc

count 5

reset

a counter

switch

state true

a flag

setTodo

last_todo ...

cancelTodo

next_todo check_plant

an agenda

...

GetLastTradePrice

a Stock Quote Web Service

availablestate

...wsdl

postEvent

registerForEvs

clearEvents

an event service

query

createTable

addRecord

a data-base

...

1001n_records

table_names ...

... ...

in

rd

out

a tuple space

Actions and Percepts in Artifact-Based
Environments [Ricci et al., 2010b]

actions←→ artifacts’ operation

the action repertoire is given by the dynamic set of operations
provided by the overall set of artifacts available in the workspace
can be changed by creating/disposing artifacts
▶ action success/failure semantics is defined by operation

semantics

percepts←→ artifacts’ observable properties + signals

properties represent percepts about the state of the environment
signals represent percepts concerning events signalled by the
environment

74



Interaction Model: Use

op(Params)

ValuePropName

ValuePropName
...

...

AGENT

op(parms)
action

▶ Performing an action corresponds to triggering the execution
of an operation
▶ acting on artifact’s usage interface

Interaction Model: Operation execution

OPERATION EXECUTION
op(Params)

ValuePropName
Value
...

...

SIGNALS OBS PROPERTIES
CHANGE

AGENT

op(parms)
action

action completion
- with success or failure -

▶ a process structured in one or multiple transactional steps
▶ asynchronous with respect to agent

▶ ...which can proceed possibly reacting to percepts and
executing actions of other plans/activities

▶ operation completion causes action completion
▶ action completion events with success or failure, possibly

with action feedbacks

Interaction Model: Observation

Belief base
(or alike)

PropName(Value).
PropName(Value).
... ValuePropName

ValuePropName
...

focus

AGENT
OBSERVER

▶ Agents can dynamically select which artifacts to observe
▶ predefined focus/stopFocus actions

Interaction Model: Observation

AGENT
OBSERVER

Belief base
(or alike)

PropName(Value).
PropName(Value).
... ValuePropName

ValuePropName
...

use

▶ By focussing an artifact
▶ observable properties are mapped into agent dynamic

knowledge about the state of the world, as percepts
▶ e.g. belief base

▶ signals are mapped as percepts related to observable events

CArtAgO

▶ Common ARtifact infrastructure for AGent Open
environment (CArtAgO) [Ricci et al., 2009a]

▶ Computational framework / infrastructure to implement and
run artifact-based environment [Ricci et al., 2007]
▶ Java-based programming model for defining artifacts
▶ set of basic API for agent platforms to work within

artifact-based environment
▶ Distributed and open MAS

▶ workspaces distributed on Internet nodes
▶ agents can join and work in multiple workspace at a time

▶ Role-Based Access Control (RBAC) security model
▶ Open-source technology

▶ available at https://github.com/CArtAgO-lang/cartago

Example 1: A Simple Counter Artifact

class Counter extends Artifact {
  
  void init(){
    defineObsProp("count",0);
  }
  
  @OPERATION void inc(){
    ObsProperty p = getObsProperty("count");
    p.updateValue(p.intValue() + 1);
    signal("tick");
  }
}

inc

count 5

▶ Some API spots
▶ Artifact base class
▶ @OPERATION annotation to mark artifact’s operations
▶ set of primitives to work define/update/.. observable

properties
▶ signal primitive to generate signals

Example 1: User and Observer Agents

!create_and_use.

+!create_and_use : true 
  <- !setupTool(Id);
     // use
     inc;
     // second use specifying the Id
     inc [artifact_id(Id)].

// create the tool
+!setupTool(C): true 
  <- makeArtifact("c0","Counter",C).

!observe.

+!observe : true 
  <- ?myTool(C);  // discover the tool
     focus(C).

+count(V) 
  <- println(“observed new value: “,V).

+tick [artifact_name(Id,”c0”)]  
  <- println(“perceived a tick”).

+?myTool(CounterId): true 
  <- lookupArtifact(“c0”,CounterId).

-?myTool(CounterId): true 
  <- .wait(10); 
     ?myTool(CounterId).

OBSERVER(S)USER(S)

▶ Working with the shared counter

Action Execution & Blocking Behaviour

▶ Given the action/operation map, by executing an action the
intention/activity is suspended until the corresponding
operation has completed or failed
▶ action completion events generated by the environment and

automatically processed by the agent/environment platform
bridge

▶ no need of explicit observation and reasoning by agents to
know if an action succeeded

▶ However the agent execution cycle is not blocked!
▶ the agent can continue to process percepts and possibly

execute actions of other intentions

Wrap-up

▶ Environment programming
▶ environment as a programmable part of the MAS
▶ encapsulating and modularising functionalities useful for

agents’ work
▶ Artifact-based environments

▶ artifacts as first-class abstraction to design and program
complex software environments
▶ usage interface, observable properties / events, linkability

▶ artifacts as first-order entities for agents
▶ interaction based on use and observation
▶ agents dynamically co-constructing, evolving, adapting their

world

▶ CArtAgO computational framework
▶ programming and executing artifact-based environments
▶ integration with heterogeneous agent platforms

Organisation Oriented
Programming
— OOP —



?

environment



Introduction: Some definitions

▶ Organisations are structured, patterned systems of activity,
knowledge, culture, memory, history, and capabilities that are
distinct from any single agent [Gasser, 2001]
⇝ organisations are supra-individual phenomena

▶ A decision and communication schema which is applied to a set
of actors that together fulfill a set of tasks in order to satisfy goals
while guarantying a global coherent state [Malone, 1999]
⇝ definition by the designer, or by actors, to achieve a purpose

▶ An organisation is characterised by: a division of tasks, a
distribution of roles, authority systems, communication systems,
contribution-retribution systems [Bernoux, 1985]
⇝ pattern of predefined cooperation

▶ An arrangement of relationships between components, which
results into an entity, a system, that has unknown skills at the level
of the individuals [Morin, 1977]
⇝ pattern of emergent cooperation

Organisation in MAS – a definition

▶ Pattern of agent cooperation
▶ with a purpose
▶ supra-agent
▶ emergent or
▶ predefined (by designer or agents)

Organisation Oriented Programming (OOP)

Organisation 
Entity

Organisation
Specification

Agent

Agent

Agent

▶ Programming outside
the agents

▶ Using organisational
concepts

▶ To define a cooperative
pattern

▶ Program = Specification
▶ By changing the

specification, we can
change the MAS overall
behaviour

Organisation Oriented Programming (OOP)

Organisation 
Entity

Organisation
Specification

Agent

Agent

Agent First approach
▶ Agents read the program

and follow it

Organisation Oriented Programming (OOP)

Organisation 
Entity

Organisation
Specification

Agent

Agent

Agent

Second approach
▶ Agents are forced to

follow the program
▶ Agents are rewarded if

they follow the program
▶ ...

Organisation Oriented Programming (OOP)

Organisation 
Entity

Organisation
Specification

Agent

Agent

Agent

Second approach
▶ Agents are forced to

follow the program
▶ Agents are rewarded if

they follow the program
▶ ...

Organisation Oriented Programming (OOP)

Organisation 
Entity

Organisation
Specification

Agent

Agent

Agent

Components
▶ Programming language

(OML)
▶ Platform (OMI)
▶ Integration to agent

architectures and to
environment

Motivations for OOP:
Applications point of view

▶ Current applications show an increase in
▶ Number of agents
▶ Duration and repetitiveness of agent activities
▶ Heterogeneity of the agents
▶ Number of designers of agents
▶ Agent ability to act and decide
▶ Openness, scalability, dynamism

▶ More and more applications require the integration of
human communities and technological communities
(ubiquitous and pervasive computing), building connected
communities (ICities) in which agents act on behalf of users
▶ Trust, security, ..., flexibility, adaptation

Motivations for OOP:
Normative point of view

▶ MAS have two properties which seem contradictory:
▶ a global purpose
▶ autonomous agents
⇝ While the autonomy of the agents is essential, it may cause

loss in the global coherence of the system and achievement
of the global purpose

▶ Embedding norms within the organisation of an MAS is a
way to constrain the agents’ behaviour towards the global
purposes of the organisation, while explicitly addressing the
autonomy of the agents within the organisation
⇝ Normative organisation

e.g. when an agent adopts a role, it adopts a set of behavioural
constraints that support the global purpose of the
organisation.
It may decide to obey or disobey these constraints

Some OOP approaches

▶ AGR/Madkit [Ferber and Gutknecht, 1998]
▶ STEAM/Teamcore [Tambe, 1997]
▶ ISLANDER/AMELI [Esteva et al., 2004]
▶ Opera/Operetta [Dignum and Aldewereld, 2010]
▶ PopOrg [Rocha Costa and Dimuro, 2009]
▶ 2OPL [Dastani et al., 2009]
▶ THOMAS [Criado et al., 2011],
▶ ...

Moise Framework

▶ OML (language)
▶ Tag-based language

(issued fromMoise [Hannoun et al., 2000],
Moise+ [Hübner et al., 2002],
MoiseInst [Gâteau et al., 2005])

▶ OMI (infrastructure)
▶ developed as an artifact-based working environment

(ORA4MAS [Hübner et al., 2009] based on CArtAgO nodes,
refactoring of S-Moise+ [Hübner et al., 2006] and
Synai [Gâteau et al., 2005])

▶ Integrations
▶ Agents and Environment (c4Jason, c4Jadex

[Ricci et al., 2009b])
▶ Environment and Organisation ([Piunti et al., 2009])
▶ Agents and Organisation (J -Moise+ [Hübner et al., 2007])

Moise OML meta-model (partial view)

Agent Goal

MissionRole

Group

Social Scheme

create
delete

adopt
leave

create
delete

agent's actionscomposition
association

Cardinalities are not represented

concept mapping

Norm

Goal

commit 
leave

achieve

Structural
Specification

Normative
Specification

Functional
Specification

Moise OML

▶ OML for defining organisation specification and
organisation entity

▶ Three independent dimensions [Hübner et al., 2007]
(⇝ well adapted for the reorganisation concerns):
▶ Structural: Roles, Groups
▶ Functional: Goals, Missions, Schemes
▶ Normative: Norms (obligations, permissions, interdictions)

▶ Abstract description of the organisation for
▶ the designers
▶ the agents

⇝ J -Moise+ [Hübner et al., 2007]
▶ the Organisation Management Infrastructure

⇝ ORA4MAS [Hübner et al., 2009]

Structural Specification

▶ Specifies the structure of an MAS along three levels:
▶ Individual with Role
▶ Social with Link
▶ Collective with Group

▶ Components:
▶ Role: label used to assign rights and constraints on the

behavior of agents playing it
▶ Link: relation between roles that directly constrains the

agents in their interaction with the other agents playing the
corresponding roles

▶ Group: set of links, roles, compatibility relations used to
define a shared context for agents playing roles in it

Structural Specification Example

Graphical representation of structural specification of 3-5-2 Joj Team

Functional Specification

▶ Specifies the expected behaviour of an MAS in terms of
goals along two levels:
▶ Collective with Scheme
▶ Individual with Mission

▶ Components:
▶ Goals:

▶ Achievement goal (default type). Goals of this type should be
declared as satisfied by the agents committed to them, when
achieved

▶ Maintenance goal. Goals of this type are not satisfied at a
precise moment but are pursued while the scheme is running.
The agents committed to them do not need to declare that
they are satisfied

▶ Scheme: global goal decomposition tree assigned to a group
▶ Any scheme has a root goal that is decomposed into subgoals

▶ Missions: set of coherent goals assigned to roles within norms

Functional Specification Example

score a goal

m1

go towards the opponent field

m1, m2, m3

get the ball

be placed in the middle field 

be placed in the opponent goal area
kick the ball to (agent committed to m2)

go to the opponent back line

kick the ball to the goal area

shot at the opponent’s goal

m1

m1

m2 m2

m2

m3

m3

Key

goal
missions

success rate parallelismchoicesequence

Scheme

Organizational Entity

Lucio

Cafu

Rivaldo

m1

m2

m3

Graphical representation of social scheme “side_attack” for joj team

Normative Specification

▶ Explicit relation between the functional and structural
specifications

▶ Permissions and obligations to commit to missions in the
context of a role

▶ Makes explicit the normative dimension of a role

Norm Specification – example

role deontic mission TTF

back obl iged m1 get the ball, go ... 1 minute
left obl iged m2 be placed at ..., kick ... 3 minute
right obl iged m2 1 day

attacker obl iged m3 kick to the goal, ... 30 seconds

Organisational Entity

norrmative 
groups 

roles 

structural 

schemas 

missions 

functional 

group 
instances role 

player 

schema 
instances 

mission 
player 

agents 

purpose 

Organisation  
specification 

Organisation 
Entity 

links norms 

Organisation Entity Dynamics

1. Organisation is created (by the agents)
▶ instances of groups
▶ instances of schemes

2. Agents enter into groups adopting roles
3. Groups become responsible for schemes

▶ Agents from the group are then obliged to commit to
missions in the scheme

4. Agents commit to missions

5. Agents fulfil mission’s goals

6. Agents leave schemes and groups

7. Schemes and groups instances are destroyed

Organisation management infrastructure (OMI)

Responsibility

▶ Managing – coordination, regulation – the agents’ execution
within organisation defined in an organisational
specification

Organisation
Program

OMI

AgentAgentAgentAgent

(e.g. MadKit, AMELI, S-Moise+, THOMAS, ...)
103



Organisational artifacts in ORA4MAS

Workspace ora4mas

Org. 
Spec.
NOPL

agent

op
link op

Scheme
Board

\\\

op
link op

Scheme
Board

\\\

op
link op

Scheme
Board

\\\

op
link op

Group
Board

\\\

op
link op

Group
Board

\\\

op
link op

Group
Board

\\\

op
link op

Workspace
Artifact

\\\

agent

agent

▶ based on A&A and
Moise

▶ agents create and handle
organisational artifacts

▶ artifacts in charge of
regimentations,
detection and evaluation
of norms compliance

▶ agents are in charge of
decisions about sanctions

▶ distributed solution

ORA4MAS – GroupBoard artifact

GroupBoard

Specification

Players

Schemes

adoptRole

leaveRole

addScheme

removeScheme

Observable Properties:
▶ specification: the

specification of the group in
the OS (an object of class
moise.os.ss.Group)

▶ players: a list of agents
playing roles in the group.
Each element of the list is a
pair (agent x role)

▶ schemes: a list of scheme
identifiers that the group is
responsible for

ORA4MAS – GroupBoard artifact

GroupBoard

Specification

Players

Schemes

adoptRole

leaveRole

addScheme

removeScheme

Operations:
▶ adoptRole(role): the agent

executing this operation
tries to adopt a role in the
group

▶ leaveRole(role)
▶ addScheme(schid): the

group starts to be
responsible for the scheme
managed by the
SchemeBoard schId

▶ removeScheme(schid)

ORA4MAS – SchemeBoard artifact

SchemeBoard

Specification

Players

Goals

Obligations

commitMission

leaveMission

goalAchieved

setGoalArgument

Groups

Observable Properties:
▶ specification: the

specification of the scheme
in the OS

▶ groups: a list of groups
responsible for the scheme

▶ players: a list of agents
committed to the scheme.
Each element of the list is a
pair (agent, mission)

▶ goals: a list with the current
state of the goals

▶ obligations: list of
obligations currently active
in the scheme

ORA4MAS – SchemeBoard artifact

SchemeBoard

Specification

Players

Goals

Obligations

commitMission

leaveMission

goalAchieved

setGoalArgument

Groups

Operations:
▶ commitMission(mission)

and leaveMission:
operations to “enter” and
“leave” the scheme

▶ goalAchieved(goal): defines
that some goal is achieved
by the agent performing the
operation

▶ setGoalArgument(goal,
argument, value): defines
the value of some goal’s
argument

Agent integration

▶ Agents can interact with organisational artifacts as with
ordinary artifacts by perception and action

⇝ Any Agent Programming Language integrated with CArtAgO
can use organisational artifacts

Agent integration provides some “internal” tools for the agents to
simplify their interaction with the organisation:
▶ maintenance of a local copy of the organisational state
▶ production of organisational events
▶ provision of organisational actions

109



Organisational actions in Jason I

Example (GroupBoard)

...
joinWorkspace("ora4mas",O4MWsp);
makeArtifact(

"auction",
"ora4mas.nopl.GroupBoard",
["auction-os.xml", auctionGroup],
GrArtId);

adoptRole(auctioneer);
focus(GrArtId);
...

Organisational actions in Jason II

Example (SchemeBoard)

...
makeArtifact(

"sch1",
"ora4mas.nopl.SchemeBoard",
["auction-os.xml", doAuction],
SchArtId);

focus(SchArtId);
addScheme(Sch);
commitMission(mAuctioneer)[artifact_id(SchArtId)];
...

Organisational perception

When an agent focus on an Organisational Artifact, the
observable properties (Java objects) are translated to beliefs with
the following predicates:
▶ specification
▶ play(agent, role, group)
▶ commitment(agent, mission, scheme)
▶ goalState(scheme, goal, list of committed agents, list of agent

that achieved the goal, state of the goal)
▶ obligation(agent,norm,goal,dead line)
▶ ....

Organisational perception – example

Handling organisational events in Jason

Whenever something changes in the organisation, the agent
architecture updates the agent belief base accordingly producing
events (belief update from perception)

Example (new agent entered the group)

+play(Ag,boss,GId) <- .send(Ag,tell,hello).

Example (change in goal state and norm violation)

+goalState(Scheme,wsecs,_,_,satisfied)
: .my_name(Me) & commitment(Me,mCol,Scheme)

<- leaveMission(mColaborator,Scheme).

+normFailure(N) <- .print("norm failure event: ", N).

Typical plans for obligations

+obligation(Ag,Norm,committed(Ag,Mission,Scheme),DeadLine)
: .my_name(Ag)

<- .print("I am obliged to commit to ",Mission);
commitMission(Mission,Scheme).

+obligation(Ag,Norm,achieved(Sch,Goal,Ag),DeadLine)
: .my_name(Ag)

<- .print("I am obliged to achieve goal ",Goal);
!Goal[scheme(Sch)];
goalAchieved(Goal,Sch).

+obligation(Ag,Norm,What,DeadLine)
: .my_name(Ag)
<- .print("I am obliged to ",What,

", but I don't know what to do!").

Summary –Moise

▶ Ensures that the agents follow some of the constraints
specified for the organisation

▶ Helps the agents to work together
▶ The organisation is interpreted at runtime, it is not

hardwired in the agents code
▶ The agents ‘handle’ the organisation (i.e. their artifacts)
▶ It is suitable for open systems as no specific agent

architecture is required

▶ All available as open source at

http://moise.souceforge.net

Conclusions



Multiagent Sytems

▶ MAS is an organisation of autonomous agents interacting
together to achieve their goals within a shared environment

▶ MAOP is a conceptual and practical tool to design and
implement distributed, complex, huge, open, .... systems

roleorg
mission

schema

ORGAMISATION
LEVEL

AGENT
LEVEL

ENDOGENOUS
ENVIRONMENT
LEVELwsp

artifact

EXOGENOUS 
ENVIRONMENT

agent

Agents

Programming actions with
▶ high level abstraction

(beliefs, plans, goals, ...)
▶ concurrent, distributed, decoupled, open, …

Environment

Programming tools for the agents
▶ high level abstraction

(workspaces, artifacts, perception, action, ...)
▶ concurrent, distributed, decoupled, open, …

Organisation

Helping the agents to live together
▶ high level abstraction

(group, roles, schemes, norms, ...)
▶ concurrent, distributed, decoupled, open, …

122



What we have learnt in this project?

▶ MAS is not only agents
▶ MAS is not only organisation
▶ MAS is not only environment
▶ MAS is not only interaction

⇝ separation of concerns

⇝ the right tool for each problem

Further Resources

▶ http://jacamo.sourceforge.net

▶ Olivier Boissier, Rafael H. Bordini,
Jomi Hübner and Alessandro Ricci
Multi-Agent Oriented Programming:
Programming Multi-Agent Systems
Using JaCaMo
MIT Press, 2020.

